
        
            
                
            
        

    



Articles


	
Prolog/Introduction [Wed, 07 Oct 20:47]


	
Prolog/Rules [Wed, 07 Oct 20:47]


	
Prolog/Recursive Rules [Wed, 07 Oct 20:47]


	
Prolog/Variables [Wed, 07 Oct 20:47]


	
Prolog/Lists [Wed, 07 Oct 20:47]


	
Prolog/Math, Functions and Equality [Wed, 07 Oct 20:47]


	
Prolog/Putting it Together [Wed, 07 Oct 20:47]


	
Prolog/Solving a Logic Puzzle [Wed, 07 Oct 20:47]


	
Prolog/Cuts and Negation [Wed, 07 Oct 20:47]


	
Prolog/Reading and Writing code [Wed, 07 Oct 20:47]


	
Prolog/Difference Lists [Wed, 07 Oct 20:47]


	
Prolog/Definite Clause Grammars [Wed, 07 Oct 20:47]


	
Prolog/Inference Engines [Wed, 07 Oct 20:47]


	
Prolog/Testing Terms [Wed, 07 Oct 20:47]


	
Prolog/Bagof, Setof and Findall [Wed, 07 Oct 20:47]


	
Prolog/Modifying the Database [Wed, 07 Oct 20:47]


	
Prolog/Input and Output [Wed, 07 Oct 20:47]


	
Prolog/Associative map [Wed, 07 Oct 20:47]


	
Prolog/Search techniques [Wed, 07 Oct 20:47]


	
Prolog/Sorting [Wed, 07 Oct 20:47]


	
Prolog/Graphics: XPCE [Wed, 07 Oct 20:47]


	
Prolog/Higher Order Programming [Wed, 07 Oct 20:47]


	
Prolog/Constraint Logic Programming [Wed, 07 Oct 20:47]


	
Prolog/Definite Clause Grammars [Wed, 07 Oct 20:47]


	
Prolog/Combining Languages [Wed, 07 Oct 20:47]








| Next | Section Menu | Main Menu | 


Prolog/Introduction

From Wikibooks, open books for an open world

< Prolog


					Jump to:					navigation, 					search


This section covers the installation of a prolog compiler, loading your first program, and querying it. It then explains how to use facts and variables in your programs and queries.

Getting Started[edit]

Before anything can be done, a prolog compiler and a text editor need to be installed on your system. A text editor will allow you to write your prolog programs and the prolog compiler (also known as the interpreter) will allow you to execute them.

Prolog Compilers

The following prolog implementations are free (at least for personal or educational use, be sure to read the terms). Simply download one and install it according to the instructions on the website:


	

	B-Prolog





	http://www.probp.com/

	A Prolog compiler for standard Prolog with extended features such as tabling and constraint solving. Supports many platforms.




	

	SWI Prolog (recommended for use with this book)





	http://www.swi-prolog.org/

	A small and robust open-source implementation that is compliant with both Prolog standards (ISO and Edinburgh) and has many extra libraries and built-in predicates. There's even a separate toolkit for creating windows and graphics, called XPCE. Supports many platforms.

	GNU Prolog





	http://www.gprolog.org/ (was: http://pauillac.inria.fr/~diaz/gnu-prolog/ )

	A relatively new open source implementation. Has support for Constraint Logic Programming, an extension of prolog.

	Visual Prolog





	http://www.visual-prolog.com/

	A complete development environment for an Object-Oriented extension of Prolog. Includes compiler, linker, text editor, graphical dialog editors, build system, debugger, large library, and much more



The following implementations aren't free:


	

	SICSTUS Prolog





	http://www.sics.se/

	Probably the most well known professional prolog implementation and development suite. ISO-compliant, many libraries and support for constraint logic programming. Free for evaluation.

	Quintus Prolog





	http://www.sics.se/quintus/

	An especially robust and dependable implementation intended for commercial use and research projects by the same company that makes SICSTUS. Free for evaluation.



Text Editors

The programs that you will write are simple text files, to be read and written with any text editor. Some prolog implementations come with their own editor, but for those that don't here's a list of text editors. These provide the basic function that are useful for writing prolog programs, such as indentation, bracket matching and some can even be adjusted to highlight the syntax of your prolog code.


	

	Crimson Editor





	http://www.crimsoneditor.com/

	A free text-editor for windows with many features.

	GNU Emacs





	http://www.gnu.org/software/emacs/emacs.html

	A free, open-source implementation of unix' classic text editor. May be difficult to understand, has lots of features.

	Vim





	http://www.vim.org/

	A free, open-source implementation of Emacs' long-standing rival.

	Textpad





	http://www.textpad.com/

	A windows text-editor with many features. Free for evaluation.

	Eclipse Prolog Plugin





	http://eclipse.ime.usp.br/projetos/grad/plugin-prolog/index.html

	A free plugin for Eclipse.



First Steps[edit]

Once you've installed your prolog implementation, it's time to write the first program and load it into the interpreter (mainly to see if everything works). Fire up your text editor and create a text file with just the following line in it:


human(john).


Be precise, capitalization is important in prolog, as is the period. This will be your program (also known as the database or the knowledge base). Give it a nice name like prolog1.pl and save it. Note: The extension pl isn't officially associated with prolog and can cause conflicts if you're also programming in Perl, which uses .pl as well. If this is a problem you can use pro or pr or anything you like just as well. Now start your prolog interpreter. Most prolog interpreters will show you a window with some startup information and then the line


?- 


with a cursor behind it. There is usually a menu for loading files into the interpreter. If there isn't, you can type the following to load your file:


consult('FILEPATH').


And press enter. Once again, be precise, no capitals and remember the dot. Replace FILEPATH with the name and directory of your file. For instance if your file is located in C:\My Documents\Prolog\prolog1.pl then use


consult('c:/my documents/prolog/prolog1.pl').


or the shorthand


['c:/my documents/prolog/prolog1.pl'].


Note that the slashes are the other way around, since the backslash (\) has special meaning in Prolog (and most other languages). If you are using a UNIX based system such as Linux, the commands may look something like this


consult('/home/yourName/prolog/prolog1.pl').
['/home/yourName/prolog/prolog1.pl'].


Your interpreter will now hopefully tell you that the file is loaded correctly. If it doesn't, consult the help file or manual of your implementations on how to consult files.

Also, you can tell prolog interpreter to load file automatically, if running it with the key -s, like this[1]:


prolog -s /home/yourName/prolog/prolog1.pl


After some information you will see


?-


To see if everything is working, type


human(john).


(don't forget the period) and press Enter. Prolog will answer with a heartfelt


Yes.


Type


human(Who).


and prolog will answer


Who = john


Press enter and prolog will end with Yes and give you back your command line.

To exit prolog, type


halt.


Syntax, Facts and Queries[edit]

The line human(john). in the previous example was a prolog sentence in the form of a predicate. This type of sentence is called a fact. Predicates consist of one word of one or more characters, all lowercase, possibly followed by a number of terms. The following are examples of valid predicates:


human(john)
father(david, john)
abc(def,ghi,jkl,m)
tree
p(a ,f ,d)


The terms (the 'words' within parentheses) can take many forms, but for now we will stick to constants. These are words, again all lowercase. The first character of both a predicate and a constant needs to be a letter. Using predicates we can add facts to a program:


human(john).
human(suzie).
human(eliza).
man(david).
man(john).
woman(suzie).
woman(eliza).
parent(david, john).
parent(john, eliza).
parent(suzie, eliza).


Note the period '.' behind each line to show that the line is over. This is very important, if you forget it, your interpreter will not understand the program. You should also be aware that the names chosen for the predicates and terms do not actually mean anything to the prolog interpreter. They're just chosen to show what meaning you have for the program. We could easily replace the word human with the word spaceship everywhere and the interpreter wouldn't know the difference.

If we load the above program into the interpreter we can run a query on it. If you type


human(john).


prolog will answer


Yes.


and if you type


woman(john).


prolog will answer


No.


This also seems fairly obvious, but it's important to see it the right way. If you ask prolog human(john)., it means you are asking prolog if this statement is true. Clearly prolog can't see from the statement whether it's true, so it consults your file. It checks all the lines in the program to see if anyone matches the statement and answers Yes if it finds one. If it doesn't, it answers No. Note that if you ask


 ?- human(david).


Prolog will answer no, because we have not added that fact to the database. This is important: if prolog can't prove something from the program, it will consider it not true. This is known as the closed world assumption.

Variables[edit]

We'll update the program with human(david), so that all people in the database are human, and either a man or a woman


 human(david).
 human(john).
 human(suzie).
 human(eliza).
 man(david).
 man(john).
 woman(suzie).
 woman(eliza).
 parent(david, john).
 parent(john, eliza).
 parent(suzie, eliza).


What we have now is still not a very expressive language. We can gain a lot more expressiveness by using variables in our query. A variable is a word, just like terms and predicates, with the exception that it starts with an uppercase letter and can have both upper and lowercase characters after that. Consider the following query


human(A).


Now, the term of the predicate is a variable. Prolog will try to bind a term to the variable. In other words, you are asking prolog what A needs to be for human(A) to be true.


?- human(A).


Prolog will answer


A = john


Which is true, because the database contains the line human(john). If you press enter, prolog will answer Yes and give you back your cursor. If you press semicolon (';') prolog will show you the rest of the possibilities


A = suzie ;
A = eliza ;
No.


After eliza, prolog answers No, indicating that there are no further possibilities. If you query prolog with more than one variable it will show you all instantiations of the variables for which the query is true:


 ?- parent(Parent, Child).

 Parent = david
 Child = john ;

 Parent = john
 Child = eliza ;

 Parent = suzie
 Child = eliza ;

 No


When prolog is asked a query with a variable it will check all lines of the program, and attempt to unify each predicate with the query. This means that it will check if the query matches the predicate when the variables are instantiated a certain way. It can unify human(A) with human(john) by making A john, but it can't unify man(A) with human(john), because the predicates don't match.

If we want to make it even more difficult for prolog we can use two predicates in our query, for instance:


 ?- human(A), parent(B,A).


Now we are asking prolog for a human A who has a parent B. The comma means and, indicating that both predicates need to be true, for the query to be true. To check this query, prolog will first find an instantiation to make the first predicate true--say it make A equal to john--and then it will try to make the second predicate true--with A equal to john. If it has found two instantiations for A and B that make both predicates true, it will return them to you. You can press Enter to end the program, or a semi-colon to see more options.

Prolog may make a choice for A, to satisfy the first predicate that doesn't work with the second. Say it chooses A = suzie to satisfy human(A); no choice for B will satisfy parent(B, suzie), so prolog will give up its choice of suzie for A, and try another name. This is called backtracking.

In the example above, prolog will first find human(david) in the program and unify A with david. To make the second predicate true, it needs to find an instantiation for parent(B, david). It can't find any, so it will look for a new instantiation of human(A). It tries the next option: A = john. Now it needs to instantiate parent(B, john). It finds B = david in the line parent(david, john) and reports back to you


A = john
B = david 


If you press semicolon it will try to find a new instantiation for the second predicate. If that fails it will try to find a new instantiation for the first predicate and so forth until it runs out of options.

There is one special variable, called the anonymous variable, for which the underscore (_) character is used. When you use this character in a query, you basically say that you don't care how this variable is instantiated, ie you don't care which term it's bound to, as long as it's bound to something. If you ask prolog


 ?- parent(A, _). 


Prolog will answer


A = david;
A = john;
A = suzie;


It will not tell you how it instantiates _. However if you ask prolog


 ?- abc(_,_,_).


This will not be true by default, prolog needs to find an instantiation for all three anonymous variables in the database, such as abc(d,e,f). Since the predicate abc isn't in the database at all, the query fails. You can use the anonymous variable in your database as well. Putting


human(_).


In your database will mean that any term, whether it already exists or not, is human. So the query


 ?- human(_).


Would be true with the above fact in the database. Here the anonymous variable is used to state a property of all objects, instead of just one. If we want to state that a specific group of objects has a certain property, we need rules. The next section deals with this.

Examples[edit]

The following program describes the public transport systems of some cities:


 transport(dresden, tram).
 transport(amsterdam, tram).
 transport(amsterdam, subway).
 transport(new_york, subway).


We can ask prolog if there is a city which has both a tram system and a subway:


 ?- transport(A, subway), transport(A, tram).
 A = amsterdam ;
 fail.


Exercises[edit]

(x) Find a Family Tree somewhere, or make one up (a real one will make it easier to check your answers). Implement part of the tree (around ten people) in a prolog program using the predicates woman/1, man/1, parent/2. The number behind the predicate describes how many arguments the predicate takes. So parent/2 describes a predicate like parent(john, mary).

You can peruse w:Category:Family_trees for a suitable family tree.

Write prolog queries for the following commands and questions. Don't worry if some people are returned more than once. We'll discover how to deal with this later on.


	List the women in the database

	List the children in the database

	List all combinations of a father and his son.

	which women have both a father and a son in the database?



Can you think of a way to display those women that do not have a father listed in the database? Can you describe what you would need to write such a query?

The answers to select excercises can be found here: Prolog/Introduction/Answers

References[edit]



	↑ Working with the SWI-Prolog, don't sure about other.

This section deals with Rules, a general way to state when a predicate is true.

Rules[edit]

So far we haven't really been doing any programming, just defining objects and their properties. The first step to programming in prolog is the use of rules. With rules we state that a predicate is true, provided that other predicates are true. A rule looks like this:


a :- b, c, d.


This says that a is true, if b, c and d are true. If we have the following database


a :- b, c, d.
b.
c.
d :- e.
e.


and we were to ask prolog if a is true


 ?- a.


prolog would take the query (or goal) a and from that generate the sub-goals b, c and d. It can see that b and c are true, directly, as they are stated as facts, but d is only true if e is true, which leads to another sub-goal, e. Since e is also a fact, prolog will answer yes.

Rules with Variables[edit]

We can use predicates with terms as well.


car(corvette) :- haswheels(corvette), haswindows(corvette).


states that a corvette is a car if it has wheels and windows. This isn't a very useful statement, as we could also simply state that a corvette is a car with


car(corvette).


We can however use variables in rules. This way we can state, for instance that anything is a car provided that it has wheels and windows:


car(A) :- haswheels(A), haswindows(A).
haswheels(corvette).
haswindows(corvette).


The first line states that for car(A) to be true, haswheels(A) and haswindows(A) need to be true. If prolog is asked whether a corvette is a car with


?- car(corvette).


it will bind A to corvette, and get the sub-goals haswheels(corvette) and haswindows(corvette). If it can prove both these goals (which it can), then car(corvette) must be true. When using variables in a program it's important to note that a variable doesn't mean anything beyond the sentence it's in. If two sentences use the variable A, it's not the same variable. For example:


bird(A) :- hasfeathers(A).
dog(A) :- barks(A).


The A in the first line simply shows that the term of the predicate bird needs to be the same as the term of the predicate hasfeathers, it has nothing to do with the A in the second line.

Now we can return to our original program and add some rules.


human(david).
human(john).
human(suzie).
human(eliza).
man(david).
man(john).
woman(suzie).
woman(eliza).
parent(david, john).
parent(john, eliza).
parent(suzie, eliza).

father(X,Y) :- parent(X,Y), man(X).
mother(X,Y) :- parent(X,Y), woman(X).


We can ask prolog who is the father of eliza:


?- father(X, eliza).

X = john ;
No


Or to list all fathers in the database and their respective children:


?- father(Father, Child).

Father = david
Child = john ;

Father = john
Child = eliza ;

No


Examples[edit]


?- mother(Mother,Child).

Mother = suzie
Child = eliza ;

No


Exercises[edit]



If you've used any other programming language, you may be familiar with recursion, which refers to a function calling itself. Prolog is particularly well suited for recursion, although in our case, it will be a predicate referring to itself, rather than a function. Go to the next section to see how and why this is done.

This section is about using rules recursively. That is, letting a rule refer to itself. This is one of the most important mechanisms in Prolog, but it can be difficult to work with at first.

Recursive rules[edit]

Consider the following program.



 parent(david, john).
 parent(jim, david).
 parent(steve, jim).
 parent(nathan, steve).

 grandparent(A, B) :- parent(A, X), parent(X, B).



It shows a male line in a family tree, and defines with a rule when someone is someone's grandparent (if he's a parent of that person's parent).

What if we wanted to define when someone is an ancestor of someone else? That is, a parent, grandparent, great grandparent and so on. If we would write rules for all these situations we could define ancestor(A,B) for this particular bloodline:



 ancestor(A,B) :- parent(A, B).
 ancestor(A,B) :- parent(A, X), parent(X, B).
 ancestor(A,B) :- parent(A, X), parent(X, Y), parent(Y, B).
 ancestor(A,B) :- parent(A, X), parent(X, Y), parent(Y, Z), parent(Z,B).



Clearly this is not the way to go. It's not elegant, a lot of work and most importantly, it still doesn't define ancestor properly. This may work for a bloodline with four generations, but if we were to add a fifth parent to the program, we'd need a new line for ancestor as well. What we need is a definition for ancestor that works for any line of parents, no matter how long.

To achieve this we need to think about the definition of an ancestor. In the first place, person P is person C's ancestor, if P is C's parent. Furthermore, we can say that person A is person C's ancestor if person A is a parent of an ancestor of C. This doesn't sound very useful, since we use the word ancestor in its own definition, but used together with the first statement, it becomes a complete definition.

For instance, if we were to ask ourselves if Steve is an ancestor of John, we would look at the definitions. Steve is certainly no parent of John, so that doesn't apply. Is Steve the parent of an ancestor of John? Steve is a parent of Jim, so the question becomes, is Jim an ancestor of John? Jim is a parent of David, so is David an ancestor of John? Here we can use the first definition, because David is a parent of John.

In prolog, the definition looks like this:



 ancestor(A, B) :- parent(A, B).
 ancestor(A, B) :- parent(A, X), ancestor(X, B).



The second rule is recursive; it uses ancestor to define ancestor. The first rule is called the stop predicate as it stops the predicate calling itself.

When faced with the query ?- ancestor(steve,john). prolog will first try line one of the definition, but won't be able to unify parent(steve, john) with any line in the database. It will then try the second line and come up with the sub-goals parent(steve, X) and ancestor(X, B). It can unify parent(steve, X) with parent(steve, jim), so prolog is left with the sub-goal ancestor(jim, john). It will continue this until it's left with the subgoal ancestor(david, john) which, using the first line of the definition, is true, because parent(david, john) is true.

Prolog is very comfortable with using recursive rules. It can very easily find all ancestors of John:



 ?- ancestor(X, john).

 X = david ;
 X = jim ;
 X = steve ;
 X = nathan ;
 No



Or all people that Nathan is an ancestor of:



 ?- ancestor(nathan, X).

 X = steve ;
 X = jim ; 
 X = david ;
 X = john ;
 No



Using recursive rules[edit]

Using recursive rules is not without its dangers, it can lead to all kinds of loops, which will usually cause stack overflows, meaning prolog has run out of memory. The following guidelines should always be kept in mind when writing recursive rules.


	Start with your stop predicate.




	Always put the non-recursive predicate before the recursive predicate. If you start with the recursive predicate prolog will try to search 'deeper' before seeing if it can stop recursing. In the example above, it wouldn't actually make a difference, but your program won't always be so clean and straight-forward.




	Avoid recursion on the left




	When writing a recursive rule, if at all possible, put the recursive rule on the right. For instance, use:




a :- b, a. 



	instead of




a :- a, b.



	This is the same principle as the previous guideline, let prolog evaluate non recursive goals first. If you recurse before you've evaluated the other sub-goals, prolog could either get stuck in infinite recursions, or do a lot of unnecessary work. Sometimes it's necessary to put the recursive element on the left, but don't do it unless you know why you're doing it.



Examples[edit]


	A classic example of a recursive function is the factorial function. You can see how this is implemented in the section Math, Functions and Equality. In most programming languages, you might learn recursion by writing a function fact(), which took a positive integer as input, and returned a positive integer. In Prolog, since we usually use predicates instead of functions, we will instead define a predicate fact(X,Y) which is equivalent to the statement, "The factorial of X is Y." Then, to find the factorial of 7, for instance, we would use the query ?- fact(7,X). Prolog will then tell us: X=5040.




	One thing that can easily be defined recursively is a list. In this context we'll consider a list a collection of things (elements) that are placed in some order. To define this recursively, we'll state that a list is a single element, that is followed by a list. To end the recursion we need a second statement, which says that a list is something that consists of an element. Together, these two definitions exactly define anything that is a list.




	Lists are actually a very common structure in prolog programs, and they're used in a recursive way. The following chapter explains how.



Exercises[edit]

A PROLOG variable can represent anything; a number, a name, a structure, an array, something as complicated as the known universe. A PROLOG program works by constraining the variables until eventually they have particular values; then telling you what the values are. A simple program might be


 X is 3+2.


and when you run it, the result will be


 X=5
 Yes.


The program might not go as far as to constrain the variables to have exact values, so you might get


 equal(A,A).   % Explains that things are equal to themselves
 X is 3+2, equal(f(X,Z),Y).
 X=5
 Y=f(5,_)
 Yes


where the '_' means that you have a variable remaining as part of the solution.

You can also get a 'Yes' result for more than one value of the variables; this is called 'nondeterminism', and is OK. If no values of the variables will make a solution, PROLOG will say 'No'.





prev: Recursive Rules next: Lists


 
						Retrieved from "https://en.wikibooks.org/w/index.php?title=Prolog/Variables&oldid=1549794"					







This article was downloaded by calibre from https://en.wikibooks.org/wiki/Prolog/Variables



 | Section Menu | Main Menu | 
| Next | Section Menu | Main Menu | Previous | 


Prolog/Lists

From Wikibooks, open books for an open world

< Prolog


					Jump to:					navigation, 					search






Contents




	1 Lists

	2 Examples

	2.1 The member predicate

	2.2 The append predicate

	2.3 Reversing a list





	3 Exercises

	4 Answers to Exercises







Lists[edit]

Any language needs a way to handle collections of objects and prolog is no exception. A list in prolog can look like this:


[a, b, c, d, e]


The brackets are the beginning and the end of the list, and the commas separate the various elements. Here all the elements are atoms, but a list can contain all sorts of elements, and different types of element can occur in the same list. The following are examples of valid lists:


[a, A, B, C, D]
[p(A,V), p(a, V)]
[[a1,a2,a3],[b1,b2,b3],[c1,c2,c3]]
[[A,B], [B,C], quu([a,b,c],[a,b,d,e,f],_), c, m, D]


Most prolog code will not define a list explicitly, like these examples, but rather handle lists of any length, with many possible elements. To handle lists without knowing what's inside them or how long they are, you use the bar notation:


[Head|Tail]


In this notation the variable Head represents the leftmost element of the list and the variable Tail represents the rest of the list represented as another list. A head can be anything, from a predicate to another list, but the tail is always another list.

Some default library rules are provided, e.g., length, reverse, append (also, these can be defined easily, as shown at the bottom of this page). To see how these works, try the following codes:


 ?- length([1, 3, 6], What).


Prolog will answer 3.


 ?- append([a], b, Z).






 ?- append([a], [b], Z).


Observe the difference here.





 ?- append(X, [1, 2], [1, 1, 2]).






 ?- reverse([1,2], What).




The following program shows how to use this.


listsplit([H|T], H, T).


Here, [H|T] is a list, H is the head, and T is the tail. For the query


 ?- listsplit([a,b,c,d,e], a, [b,c,d,e]).


Prolog will answer yes. For the query


 
 ?- listsplit([a,b,c,d,e], A, B).


Prolog will answer:


A = a
B = [b, c, d, e] ;


The program can even be used to 'create' a list:


 ?- listsplit(List, a, [b,c,d,e]).
 List = [a, b, c, d, e] ;


here are some examples of lists, and what heads and tails they result in:




	List
	Head
	Tail



	

[a,b,c,d,e]


	

a


	

[b,c,d,e]





	

[a]


	

a


	

[] (an empty list)





	

[[a,b,c],a,b,[a,b]]


	

[a,b,c]


	

[a,b,[a,b]]








Note that the empty list [ ] cannot be split up and therefore will not unify with [H|T].

Splitting up lists can be done with more than two heads:


[H1, H2, H3|Tail]


This will split a list up into the first three elements, and the rest of the list. Note, however that this will fail if the list has less than three elements.

Now consider the following program:


 last([Elem], Elem).
 last([_|Tail], Elem) :- last(Tail, Elem).


This relation defines the last element of a list. It can be used as a program to find a list's last element:


?- last([a,b,c,d,e],X).
X = e


First, there is the stop predicate, that says that the last element of a list with one element, is that element. The second line says that the last element (Elem) of a list [_|Tail] is the last element of its tail (Tail). Since we don't care about the head, we use the anonymous variable, _, for it.

Examples[edit]

The member predicate[edit]

Normally, you would use builtin predicates for these list operations, instead of writing them yourself. Builtin predicates are defined by your prolog implementation, but can be used in any program. These implementations are shown here to illustrate how to modify lists.

Member is a standard prolog built-in predicate. You use it like this:


member(Element, List).


Where List is any prolog list, and Element is any element in that list. The following, for instance, succeeds (returns 'Yes'):


 ?- member(a, [a, b, c]).
 ?- member(b, [a, b, c]).       
 ?- member(c, [a, b, c]).


This query:


 ?- member(Element, [a, b, c]).


Will return the following values for Element:


Element = a;
Element = b;
Element = c;


The member predicate is defined like this:


member(X, [X|_]).        % member(X, [Head|Tail]) is true if X = Head 
                         % that is, if X is the head of the list
member(X, [_|Tail]) :-   % or if X is a member of Tail,
  member(X, Tail).       % ie. if member(X, Tail) is true.


The append predicate[edit]

The built-in predicate append/3 attaches a list to the back of another, in other words, it concatenates two lists. It's used like this:


append(Xs, Ys, Zs)


Where Zs is Ys appended to Xs. The following succeed:


append([a, b, c], [1, 2, 3], [a, b, c, 1, 2, 3]).
append([], [a, b, c], [a, b, c]).
append([A, B, C], [1, 2, 3], [A, B, C, 1, 2, 3]).


You can use it to append two lists:


 ?- append([a, b, c], [d, e, f], Result).

 Result = [a, b, c, d, e, f]


Or to split a list into left and right parts,


 ?- append(ListLeft, ListRight, [a, b, c]).
 
 ListLeft  = []
 ListRight = [a, b, c] ;
 
 ListLeft  = [a]
 ListRight = [b, c] ;
 
 ListLeft  = [a, b]
 ListRight = [c] ;
 
 ListLeft  = [a, b, c]
 ListRight = [] ;
 
 No


You can even use it with three variables. (Try this for yourself to see what the result looks like).

The append predicate can be defined like this:


append([], Ys, Ys).
append([X|Xs], Ys, [X|Zs]) :- append(Xs, Ys, Zs).


The first line simply unifies the last two lists, it succeeds for queries like: append([], [a,b,c], [a,b,c]). or append([], X, [e]). in which case it would bind X = [e]. The second line (the recursive clause) declares that, given append(A,B,C) the head of A is equal to the head of C, and appending the tail of A, with B, gives the tail of C.

Due to the nondeterminism of Prolog, append/3 has many uses. It can be used as another way to implement last/2 which was defined earlier.


last(List, Last) :- append(_, [Last], List).


Many more definitions are possible,


split(List, Pivot, Left, Right) :- append(Left, [Pivot|Right], List).



?- split([o,o,x,e,e,e], x, L, R).
L = [o, o],
R = [e, e, e] ;



?- split(A, -, [o,o], [u,u,u]).
A = [o, o, -, u, u, u].


Reversing a list[edit]

We will look at two ways to reverse a list, first of all the naive way is to simply keep taking the head off and appending it onto the end,


reverse([],[]).
reverse([X|Xs],YsX) :- reverse(Xs,Ys), append(Ys,[X],YsX).


Executing it means that you traverse the list over and over again appending each time, A more efficient version can be created by taking the definition of append:


append([], Ys, Ys).
append([X|Xs], Ys, [X|Zs]) :- append(Xs, Ys, Zs).


and changing it slightly,


revappend([], Ys, Ys).
revappend([X|Xs], Ys, Zs) :- revappend(Xs, [X|Ys], Zs).


then,


reverse(Xs,Ys) :- revappend(Xs,[],Ys).


The strategy used in revappend is called an accumulating parameter.

Exercises[edit]

(1) The built in predicate length() can be used to find the length of a list. For example:


?- length([a,b,95,[1,1,1,1]],X).
X = 4 .
?- length([a,XYZ,59,1,1,1,1],X).
X = 7 .


How might this predicate be defined?

Answers to Exercises[edit]

(1) As you probably guessed, we will use recursion.


len([], 0).
len([_ | Tail], Length) :-
    len(Tail, Length1),
    Length is Length1 + 1,!.


Another solution which uses tail recursion optimisation and Prolog's arithmetic (And therefore uses less stack space):


% 0 - Calling Rule
cl(List, Out) :-
        call(List, 0 , Out).

% 1 - Terminating condition
call([], Count, Count).

% 2 - Recursive rule
call([H|T], Count, Out) :-
        Count1 is Count + 1,
        call(T, Count1, Out).

This section explains how to use math in prolog, using functions and equality.

Numbers in Prolog[edit]

Prolog, like any other programming language, has a representation for numbers. Numbers are used like constants, and represented like they are anywhere on the computer, the following are valid ways of dealing with numbers in a predicate:


a(12, 345).
a(A) :- b(A, 234.3).
a(345.3409857) :- b(23476.923804).


To perform mathematical operations on numbers, we will need functions. To store the result of a mathematical operation in a variable, we will need to look more closely at equality.

Functions[edit]

Until now, predicates have always represented a simple true or false. Predicate a(A, B) is true or false, depending on the values of A and B. Functions are predicates that represent a value. The sin() predicate, for instance, is a function. sin(0) represents the value 0 and sin(1) represents the value 0.841471. Functions can be used anywhere a number or constant can be used, in queries, predicates and rules. For instance, if the fact p(0). is in your program, the query ?- p(sin(0)). will unify with it.

The following common mathematical functions are built in to most Prolog implementations:



	function
	example
	result



	+
	2 + 3
	5



	-
	4 - 1
	3



	*
	4 * 3
	12



	/
	22/7
	3.14286



	^
	4 ^ 2
	16



	sin
	sin(3)
	0.14112




(table to be completed)

Note that functions themselves cannot be evaluated. the query ?- sin(3). will fail because sin() is implemented as function and not as a predicate.

One difference between functions and predicates is that the meaning (or definition) of a predicate is usually defined by you, in your program. When you use functions like sin(), they've already been defined in your prolog implementation. In other words, prolog will not find the definition in your program, but in it's library of built-in predicates. It is possible to create your own functions, but that's something you will usually not need.

Equality[edit]

There are several kinds of equality, with slightly different meanings. Here we will just look at the = operator and the is operator. First, look at:


?- A is 22/7.


This query assigns the result of mathematical operation 22/7 to the variable A. So Prolog will answer:


A = 3.14286


This idea may be familiar from other programming languages. The = operator, however, is very different. It doesn't solve the right-hand side, but instead keeps it as a formula. So you get this:


?- A = 22/7.
A = 22/7


Instead of assigning the result of the operation to the variable A, prolog assigns the operation to A, without evaluating it.

You can see the same thing with queries. If you ask


?- (1 is (2-1)).


You'll get "Yes", because the (2-1) is evaluated (solved). But if you ask


?- (1 = (2-1)).


You'll get "No", because Prolog will compare a number (1) to a formula (2-1), rather than to the result of solving the formula.

The is operator is meant specifically for mathematical functions. The left argument has to be a variable and the right argument has to be a mathematical function with all variables instantiated. The = operator is used for unification of variables and can be used with any two arguments (although it will fail if the two arguments aren't the same and can't be made the same by instantiating variables a certain way).

Prolog knows many other ways of comparing two terms or instantiating variables, but for now, these two will suffice. When working with functions, we will almost always use the is operator.

Math[edit]

Now that we know about functions and equality, we can start programming with math.


plus(A, B, C) :- C is A + B.


This predicate adds two numbers (A and B), and unifies the result with C. The following program is somewhat more complex.


fac(0,1).
fac(A,B) :- 
      A > 0, 
      Ax is A - 1,
      fac(Ax,Bx), 
      B is A * Bx.


This program calculates the factorial of A (A! in math notation).

It works recursively. The first rule states that the factorial of 0 is 1. The second states that the factorial of a number A greater than 0 is the factorial of A-1 times A.

Examples[edit]

Exercises[edit]

(1) What will prolog answer to the following queries (on an empty database)? Try to think of the answer yourself, and then use a prolog compiler to verify it.


	 ?- X = 1 + 2 + 3.

	 ?- X is 100/10.

	 ?- X is (14 + 16)/3, X + 3 = Y.

	 ?- X = 1000/100 + 5, Y is X.



(2) Write a predicate called sigma, such that sigma(A,B,N) is true when N=A+(A+1)+(A+2)+...+(B-2)+(B-1)+B. In other words, [image: \sum_{i=A}^B i = N]. You may assume that A and B are integers with B>A. Test your predicate with queries such as:


?- sigma(4,9,X).
X = 39 ;
fail.

?- sigma(-7,-2,X).
X = -27 ;
fail. 

?- sigma(-5,5,X).
X = 0 ;
fail.


(3) The factorial program shown at the end of this chapter sins against one of the guidelines of using recursive rules. In the second rule:


fac(A,B) :- 
      A > 0, 
      Ax is A - 1,
      fac(Ax,Bx), 
      B is A * Bx.


The recursive part is not the last predicate in the rule.


	Show how prolog evaluates the query ?- fac(3, X). and explain why the program is set up like this.

	Explain why the line A > 0 is necessary. What would prolog do after it had found an answer, if the line was missing?



Answers to Exercises[edit]

(2) As usual, there is more than one way to solve this problem. Here's one way which uses recursion in a similar way to the factorial predicate.


sigma(A,A,A).
sigma(A,B,N) :-
        B>A, %What do you think would happen if you removed this line? Try it. Why does this happen?
        A1 is A+1,
        sigma(A1,B,N1),
        N is A+N1.

This section serves both to combine the knowledge from the previous chapters and to review it. Prolog is first explained in more detail, true to its basic structure, and then a small program is presented that combines the knowledge from the previous chapters.

Prolog[edit]

A Prolog program consists of a database of rules. These rules are loaded into the compiler. The user can then run queries on the database. The compiler will try to prove that the query is true.

Everything in Prolog is described in logic (Prolog logic is variant of classic predicate- or first order logic). Starting from the atoms (the smallest bits) these are the most important logical symbols that make up a Prolog program or query.

constants[edit]

These consist of one or more lowercase characters, starting with a letter. A constant is not made up of other Prolog symbols (it's an atom) and it does not represent anything in the Prolog world except itself. A good way to think of atoms is as the objects of the prolog world. They have no meaning attached to them. Examples of valid constants:


foo
cd_p0345
a
dirk


Variables[edit]

Variables are containers. They can represent pretty much any Prolog structure, from constants to predicates. Variables consist of one or more lower- or uppercase characters, starting with an uppercase letter. It's important to note that variables function differently in Prolog than in most programming languages. You do not assign a value to them and then use them, various values are assigned to them by Prolog when you run a query to try and make the query true. Examples of valid variables:


B
Parent
Result


Predicates[edit]

Predicates consist of a functor (one or more lowercase characters, starting with a letter), possibly followed by a number of terms between brackets, separated by commas. A predicate is either true or false, based on its terms. Examples of valid predicates:


childof(Child, Father)
predicate(dog(x), cat(Y))
architect(a_vandelay)
termless_predicate


Functions[edit]

A function is built up the same way as a predicate, but instead of being false or true, it can represent anything a variable can represent. For instance, the function sqrt(A), will return the square root of whatever number A represents.

Atomic Sentences[edit]

An atomic sentence is the smallest part in a prolog program that can take on the meaning true or false. All predicates are atomic sentences. Some other Prolog statements, such as unification (=) are atomic sentences as well. Examples of atomic sentences:


childof(george_w_b, george_b)
A = tan(B, C) + D


Sentences (and connectives)[edit]

Sentences are Prolog constructions that can take on the meaning true or false. Atomic Sentences are of course sentences. Other sentences (i.e., non-atomic ones) are comprised of atomic sentences joined together by connectives. The most important connective in Prolog is , (the "and" connective). If a sentence joins two atomic sentences with a comma, the sentence will be true if (and only if) both atomic sentences are true. Most sentences in Prolog are just a number of predicates with comma's between them, indicating that all atomic sentences need to be true for this sentence to be true. Note: The other connectives, important though they are, will come up later. Examples of sentences.


childof(a, b), male(a), female(b)
A = B * 3, even(A)
dog(poochie), cat(scratchy), mouse(itchy)


Rules[edit]

A rule is a special type of sentence. It has a predicate on the left, and a sentence on the right separated by :-. It's concluded by a period. Sentences are the lines in a Prolog program. A rule states that the predicate is true if the sentence is true. If there are variables in the rule, then the predicate is true for any instantiation of the variables for which the sentence is true. A rule with no sentence after the predicate (and thus, no :-) is called a fact and is considered true as is. Every sentence in a Prolog program is a rule.

Examples of rules.


mother(A) :- has_child(A), female(A).
car(porsche).
equal(X, X).


Terms[edit]

A term is any Prolog construct representing an object. Constants and functions are always terms and variables can represent terms.

Other Prolog concepts are explained in the Glossary.

The Prolog Compiler[edit]

The Prolog compiler allows users to run queries on a database of rules (a Prolog program). Queries are special types of rules, with no predicate before the :-. Once a query is entered, Prolog will check if it's true, given that the rules in the database are true. If the query has variables, Prolog will attempt to find an instantiation of the variables that will make the query true.

This instantiating of variables is based on the process of unification. If Prolog is trying to verify the query :- a(X). and encounters the rule a(a). in the database, it will unify the variable X with the constant a. In other words, it will instantiate the variable X to the value a. Since a(a) is true, Prolog will return X = a as a solution. Unification takes two predicates and sees if one can be used to instantiate the variables of the other. The following two predicates unify:


p(A, y)
p(z, B)
unification: A = z, B = y


As do these:


q(A, [y,z,C])
q(x, B)
unification: A = x, B = [y,z,C]


However, these will not unify:


q(A, A)
q(x, y)


This is because A can be x or y, but not both. These on the other hand:


q(A, B)
q(x, x)


will unify, because two different variables can be bound to the same constant.

If Prolog is trying to verify the query :- b(X). and encounters the rule b(A) :- c(A). It will unify the variable X to the variable A. Neither of the variables are instantiated yet (i.e., have values assigned to them) but they are bound to each other, they can only be instantiated to the same value. Prolog will now add c(A) to its list of queries to verify. If it encounters the rule c(c) :- d. in the database, it will (temporarily) instantiate A with c, and try to verify d. If d succeeds, Prolog will return X = c as a solution to the original query, if Prolog can't find anything in the database that makes d true, Prolog will forget this rule and try to make c(A) true some other way. This is known as backtracking.

An example program[edit]

The following program checks if the sum of a list of numbers equals zero:


sum_is_zero(List) :- 
        sum(List, Sum),                 % Sum is the sum of the List
        Sum = 0.                        % Sum needs to be 0


sum([], 0).                             % the sum of an empty list is always 0 (stop predicate)

sum([FirstNumber|Tail], Sum) :-            % The list is split into its first number and its tail
        sum(Tail, TailSum),             % TailSum is the sum of the tail
        Sum is FirstNumber + TailSum.   % Add the first number to that to get the Sum of the list.


The first rule in the program is pretty straightforward. It sums the list, using the sum predicate, and checks if it's equal to zero. The second two rules constitute the sum predicate. The first rule is the stop predicate. It simply states that an empty list has a sum of 0. The second rule first splits its list into a FirstNumber (the first number in the list, the 'head') and a Tail (the rest of the list). It calculates the sum of the tail recursively, and adds the first number to the sum of the tail.



Some queries:


?- sum_is_zero([1, -1]).
Yes

?- sum_is_zero([1, 1, -2]).
Yes

?- sum([-1.5, 2.7849, 3.383724], A).
A = 4.66862 ;
No

?- sum([-1.5, 2.7849, 3.383724], 42).
No


Examples[edit]

Exercises[edit]

(1) Which of the following predicate pairs unify? If they do, give the instantiation of the variables, if they don't, explain why not.


1 child_of(M, P)
  child_of(martha, doug)
2 dog(A)
  dog(B)
3 f(X, Y, e, Z)
  f(a, b, d, c)
4 r(D, F)
  r(p, c(martha, D))

Let's work with a fun logic problem. You can work out the solution in your head or on scratch paper, then we'll solve it using Prolog.

Consider a group of ten friends who want to visit a new city somewhere in the world. They vote on seven potential destinations:


	Cairo

	London

	Beijing

	Moscow

	Mumbai

	Nairobi

	Jakarta



One city received four votes, two cities received two votes each, two cities received one vote each, and the remaining two cities received zero votes. How many votes did each of the cities receive?


	Beijing and Cairo got different numbers of votes.

	Moscow either got the most votes, or it got zero votes.

	Cairo got more votes than Jakarta did.

	In the list of cities above, each of the two cities that got two votes has a city that got no votes immediately above it in the list.

	Either Jakarta got one fewer votes than London did, or it got one fewer vote than Beijing did.

Solution in Prolog[edit]

Here is one possible solution in Prolog. We'll use the city names to represent the number of votes each received. With the generate-and-test paradigm that Prolog encourages, we'll generate all permutations of votes and test them against the puzzle rules.

Distribution of votes[edit]

One city received four votes, two cities received two votes each, two cities received one vote each, and the remaining two cities received zero votes. We can use the built-in permutation predicate.



permutation([Cairo, London, Beijing, Moscow, Mumbai, Nairobi, Jakarta],[4,2,2,1,1,0,0])



Rule 1[edit]

Beijing and Cairo got different numbers of votes. We can use built-in comparison operators. Recall that the semi-colon means OR.



(Cairo < Beijing; Cairo > Beijing)



Rule 2[edit]

Moscow either got the most votes, or it got zero votes.



(Moscow = 4; Moscow = 0)



Rule 3[edit]

Cairo got more votes than Jakarta did. This is a straight comparison operation.



(Cairo > Jakarta)



Rule 4[edit]

In the list of cities above, each of the two cities that got two votes has a city that got no votes immediately above it in the list. This rule is tougher to solve. You might consider modifying the built-in member predicate to look for pairs. Here instead we define a count predicate to count the number of times a pair appears in a list. Then we look for a list in which 0,2 appears twice.



 count([],_,0).
 count([X,Y|Rest],[X,Y],N) :- count(Rest,[X,Y],N1), N is N1 + 1.
 count([Z|Rest],[X,Y],N) :- Z \= X, count(Rest, [X,Y], N).

 count([Cairo, London, Beijing, Moscow, Mumbai, Nairobi, Jakarta], [0,2], 2)



Rule 5[edit]

Either Jakarta got one fewer votes than London did, or it got one fewer vote than Beijing did. Recall that 'is' evaluates the operation.



(Jakarta is (London-1); Jakarta is (Beijing-1))



Complete solution[edit]



count([],_,0).
count([X,Y|Rest],[X,Y],N) :- count(Rest,[X,Y],N1), N is N1 + 1.
count([Z|Rest],[X,Y],N) :- Z \= X, count(Rest, [X,Y], N).

votesFor([Cairo, London, Beijing, Moscow, Mumbai, Nairobi, Jakarta]) :-
    permutation([Cairo, London, Beijing, Moscow, Mumbai, Nairobi, Jakarta],[4,2,2,1,1,0,0]),
    (Cairo < Beijing; Cairo > Beijing),
    (Moscow = 4; Moscow = 0),
    (Cairo > Jakarta),
    count([Cairo, London, Beijing, Moscow, Mumbai, Nairobi, Jakarta], [0,2], 2),
    (Jakarta is (London-1); Jakarta is (Beijing-1)).



Prolog returns the following values:



?- votesFor([Cairo, London, Beijing, Moscow, Mumbai, Nairobi, Jakarta]).
Cairo = 4,
London = Moscow, Moscow = 0,
Beijing = Mumbai, Mumbai = 2,
Nairobi = Jakarta, Jakarta = 1 .



In fact, Prolog returns the solution above eight times. That's because two cities share a vote total in three separate places (2^3 = 8), and Prolog finds each different way of arriving at the same solution.

This section explains the use of cuts to control backtracking, the way negation is used in Prolog and a useful combination of both.

The cut[edit]

Consider the following program.



a :- b.
b :- c.
c :- d.
b.



If you load this into Prolog and ask Prolog ?- a., Prolog will evaluate it by first searching for any rule that can make a true; it adds a to its list of goals to prove. The first rule states that a is true if b is true, so Prolog adds the goal b to its list. It finds a rule for c and now needs to prove d, which fails. Prolog now removes the goal d from its list, because it couldn't prove it, and tries to prove c in a different way. This is known as backtracking. Prolog can't prove c either, backtracks again and tries b. It can prove this, using the last line of the program and Prolog terminates.

Understanding how Prolog evaluates your query is essential in Prolog programming. To control the way Prolog evaluates your program, you can use the cut operator: !. the cut operator is an atom, and can be used in the following way:



a(X) :- b(X), c(X), !, d(X).



If Prolog finds a cut in a rule, it will not backtrack on the choices it has made. For instance, if it has chosen frank for the variable X and encounters a cut, Prolog will consider frank the only option for X, even if there are other possibilities in the database. This is illustrated by the following program



a(X, Y) :- b(X), !, c(Y).
b(1).
b(2).
b(3).

c(1).
c(2).
c(3).



If we ask Prolog ?- a(Q, R). it will first answer



?- a(Q, R).

Q = 1

R = 1 ;


And when we ask Prolog for more answers, using the ;-key: 

Q = 1

R = 2 ;

Q = 1

R = 3 ;

No
.


As you can see Prolog considers 1 as the only option for Q, whereas it returns all alternatives for R. When Prolog starts out on the query it tries to prove a(Q, R), using the first line of the program. To prove this rule, it needs to first prove b(Q), it succeeds with Q = 1. Then Prolog encounters a cut and sets Q = 1 as the only option for Q. It continues with the last goal of the rule, c(R). It first finds R = 1. and completes its goal. When user presses ;, Prolog first checks for alternatives to the goal c(R). Remember, when Prolog encountered the cut it hadn't chosen an instantiation for R yet, so it can still look for alternatives for R. When it runs out of alternatives for R, Prolog can't look for alternatives for Q, and terminates.

To understand how the cut changes the meaning of a program, consider the following:




a(X) :- b(X), !, c(X).
b(1).
b(2).
b(3).

c(2).



Without the cut in the first line, Prolog would return Q = 2 to the query ?- a(Q). With the cut Prolog fails to find an answer. To prove the goal, it first proves b(Q) with Q = 1. It then encounters a cut, is committed to Q=1 and can't find a proof for c(1). If it could have searched for alternatives, it would have found that Q=2 makes both b(Q) and c(Q) true, but the cut doesn't allow that. However, if we ask Prolog a(2), it will return yes. Now, Prolog instantiates the X in the first line of the program with 2, because the user has specified it, and when Prolog reaches the cut X becomes committed to 2, allowing Prolog to prove c(2).

The cut also works across multiple rules. If a program has two rules for the goal a(X), and Prolog encounters a cut in that rule, it is not only committed to the instantiations of the variables, but also to that rule for a(X). Prolog will not consider the second rule. For instance, the following program:



a(X) :- b(X), !, c(X).
a(X) :- d(X).

b(1).
b(4).

c(3).

d(4).



will fail for the query ?- a(X).. Prolog could solve the query with the second rule, using X =4 and the last line of the program, but Prolog tries the first rule first and when it encounters the cut, it is forced to ignore all alternatives to a(Q). This time, the query ?- a(4) will fail too, because Prolog still reaches the cut. When it tries a(4) with the first rule, it succeeds in proving b(4) and reaches the cut. It then tries c(4), which fails, and Prolog has to terminate. If the lines b(4). and b(1). are removed from the program, Prolog fails on the first rule, before it encounters the cut, and is allowed to solve the query with the second rule.

Using Cuts[edit]


	examples where cuts are necessary

	if-then-else structures

	advice on using cuts (red cuts & green cuts)



Negation[edit]

not(X) is the way to implement negation in Prolog; however not(X) does not mean that X is false, it means that X can't be proven true.

For example, with the database:



man('Adam').
woman('Eve').



asking not(man('Abel')). will return yes.

Cut-Fail negation[edit]

Examples[edit]



not(Goal) :- call(Goal),!,fail.
not(Goal).



Exercises[edit]





previous: Putting it Together next: Reading and Writing code

This section gives guidelines and tips on reading and writing prolog code, and how to think about prolog code in general.

Examples[edit]

fac(0,1).

fac(A,B) :- A > 0, C is A-1, fac(C,D), B is A*D.

Exercises[edit]





previous: Cuts and Negation next: Difference Lists

Definition[edit]

A Difference list in Prolog is a normal list except the very end of it is a logic variable, paired with that variable. For example:


 [a,b,c|E]-E


Demonstration[edit]

A common example to explain why they are useful is:


 append(I-M, M-O, I-O).


Given this clause one can query:


 ?- append([a,b,c|E]-E,  [x,y,z|W]-W,  O).
 E = [x, y, z|W],
 O = [a, b, c, x, y, z|W]-W.


Since this uses a single unification to append you have O(1) instead of O(n) complexity.

Internal to Definite Clause Grammars[edit]

You can expect a DCG to expand into normal Prolog rules which use difference lists, for example.


?- expand_term(( o --> [a,b,c] ), E).
E = (o(I, O) :- I=[a, b, c|O]).


is a valid expansion.

Examples[edit]

One of the earlier examples, revappend, can be written using DCG:


revappend([]) --> [].
revappend([X|Xs]) --> revappend(Xs), [X].

Prolog has a mechanism for defining certain grammars called Definite Clause Grammar notation. This makes it easy to write parsers. Note that while DCG syntax is part of the ISO standard, DCG semantics are not.[citation needed]

Grammar rules can be written in the form:


head --> body


For example:


sentence --> noun_phrase, verb_phrase.


Which will be translated to:


sentence(L0,LREMAINDER):-
   noun_phrase(L0,L1),verb_phrase(L1,LREMAINDER).


It means, that the sentence clause will receive L0 as input, and after parsing from a sentence from L0, it will give back LREMAINDER. Let's assume that your start symbol is sentence. Then LREMAINDER is expected to be [ ] after a succesful parsing. The interpretation of the body of this clause is: if we parse a noun_phrase and a verb_phrase from our sentence, we will get back an empty list.

You can also call prolog predicates using braces.

Example[edit]

An example DCG program, which can parse numbers:


number --> digit, number_remaining.
number_remaining --> dot,number_remaining.
number_remaining --> digit,number_remaining.
number_remaining([],[]).
dot -->[0'.].
digit --> [J], {digit_code(J)}.
digit_code(J):- J >= 0'0, J =< 0'9.


Note: 0'9 means the character 9 (or more precisely, the character code of 9, because there's no distinct character datatype in SWI).

If you try to consult this program, you might get a warning, because we redefined the built-in number/2 predicate.

Running the program:


?- number("120",L).
L = [] ;
fail.


We get back one "result": it means that the parsing wasn't ambiguous. (There's only one possible parsing tree of the input.)

One of the earlier examples, revappend, can be written using DCG:


revappend([]) --> [].
revappend([X|Xs]) --> revappend(Xs), [X].


External links[edit]

Chomsky hierarchy of grammars on Wikipedia

This article needs attention from experts!

By default Prolog does a top-down logic inference.

Example:



clever(joe).
handsome(joe).

hasgirlfriend(X):-
    clever(X),handsome(X).



We run this as "hasgirlfriend(X)." to get the people who have a girlfriend.

Bi-directional inference:



clever(joe).
handsome(joe).

rule([handsome(joe),clever(joe)],(hasgirlfriend(joe))).

run:- call(rule(X,Y)),findall(A,(member(A,X),call(A)),L),
	length(X,Rulelength),
	length(L,Rulelength),
	assert(Y),print(Y).



We run this as "run.".

We don't have to directly ask who has a girlfriend, it gives us all results that can be implied. The benefit of bidirectional inferencing is the speed gain.

process(File) :-
        open(File, read, In),
        get_char(In, Char1),
        process_stream(Char1, In),
        close(In).
 
process_stream(end_of_file, _) :- !.
process_stream(Char, In) :-
        print(Char),
        get_char(In, Char2),
        process_stream(Char2, In).

Bagof, Setof and Findall are so called metapredicates, because they take a ":Goal" as an argument. Metapredicates are the equivalent of higher order functions from functional programming languages.

What is a :Goal?[edit]

A :Goal is anything that you can feed in to the top level interpreter. For example the simplest metapredicate is call/1.

Call/1 is similar to "eval" in other scripting languages.

We mark goals with the prefix ":" in header comments. It's just a convention we use in documentation, and comments. The Prolog interpreter knows nothing about this convention.

The header comment for call/1 is:


call(:Goal)


Examples:


 ?- call(write('hello')).
 hello
 true.

 ?- call(A=1).
 A = 1.

 ?- X=write('hello'), call(X).
 hello
 X = write(hello).


From the SICSTUS manual:


 call(:Term) [ISO]
 :Term
 If Term is instantiated to a term which would be acceptable as the body of a clause, then the goal call(Term) is  executed 
 exactly as if that term appeared textually in its place, except that any cut (!) occurring in Term only cuts alternatives
 in the execution of Term.


All metapredicates call "call/1" on one way or other.

Finding all the solutions without pushing ";" all the time[edit]

Sometimes we want to restrict the standard Prolog backtracking to a block of code, and put all solutions that the backtracking found in to a list that we can use outside of that block. That's when we use "findall/3".

Example: (SWI prolog)

We need modulo division for this example. "mod/2" works the following way:


 ?- A is mod(5,2).
 A = 1.

 ?- A is mod(4,2).
 A = 0.


We also need numlist/3, which simply generates a list of successing integers.


 ?- numlist(1,8,X).
 X = [1, 2, 3, 4, 5, 6, 7, 8].


Now comes the meat: we want to filter the odd numbers from the above list. We use findall/3 for this:


findall(+Template, :Goal, -Bag)



?- findall(X, (numlist(1,8,NL),member(X,NL),0 =:= mod(X,2)) ,L).
L = [2, 4, 6, 8].


As you see, the middle argument is a simple goal. If copy it and feed it to the toplevel interpreter you get the following:


 ?- numlist(1,8,NL),member(X,NL),0 =:= mod(X,2).
 NL = [1, 2, 3, 4, 5, 6, 7, 8],
 X = 2 ;
 NL = [1, 2, 3, 4, 5, 6, 7, 8],
 X = 4 ;
 NL = [1, 2, 3, 4, 5, 6, 7, 8],
 X = 6 ;
 NL = [1, 2, 3, 4, 5, 6, 7, 8],
 X = 8 ;
 false.


The first argument of findall/3 denotes the variable we want to collect in the list in the third argument.

You can read the above example as: find all X where X statisfies :Goal.

Bagof is very similar to this, but you can also use the existential quantifier "^" in the +Template. The variable on which the existential quantifier is applied won't be collected in the resulting list.

Setof is like bagof but the resulting list is ordered and doesn't contain repetitions.

Further examples[edit]

Find divisor pairs:


?- findall(X-Y, (numlist(1,8,NL),member(X,NL),member(Y,NL),X>Y,Y =\=1, 0 =:= mod(X,Y)), L).
L = [4-2, 6-2, 6-3, 8-2, 8-4].


The same, but we don't care about Y.


?- bagof(X, Y^(numlist(1,8,NL),member(X,NL),member(Y,NL),X>Y,Y =\=1, 0 =:= mod(X,Y)), L).
NL = [1, 2, 3, 4, 5, 6, 7, 8],
L = [4, 6, 6, 8, 8].


Filtering out repetition:


?- setof(X, Y^(numlist(1,8,NL),member(X,NL),member(Y,NL),X>Y,Y =\=1, 0 =:= mod(X,Y)), L).
NL = [1, 2, 3, 4, 5, 6, 7, 8],
L = [4, 6, 8].

This page is a stub.

From the SWI manual: [1]


asserta(+Term)
Assert a fact or clause in the database. Term is asserted as the first fact or clause of the corresponding predicate. Equivalent to assert/1, but Term is asserted as first clause or fact of the predicate.

assertz(+Term, -Reference)
Equivalent to asserta/1, asserting the new clause as the last clause of the predicate.

retract(+Term)
When Term is an atom or a term it is unified with the first unifying fact or clause in the database. The fact or clause is removed from the database.

retractall(+Head)
All facts or clauses in the database for which the head unifies with Head are removed. If Head refers to a predicate that is not defined, it is implicitly created as a dynamic predicate. See also dynamic/1.35



Examples:

Let's start with an empty database:


?- human(john).
ERROR: toplevel: Undefined procedure: human/1 (DWIM could not correct goal)

?- assert(human(john)).
true.

?- human(X).
X = john.

?- 

Filename convention[edit]

Filenames must be put between apostrophes, so they become atoms. ( "foo.txt" will not work, because it's a list of characters and not an atom.) In Windows systems you can write paths as 'C:/foo.txt' or 'C:\\foo.txt'.

Reading a file the Edinburgh style[edit]

You can open a file with see/1 and close with seen/0. The following program reads a file, and prints it to the standard output:






process(X):-
        X = 'c:/readtest.txt',
        see(X),
        repeat,
        get_char(T),print(T),T=end_of_file,!,seen.



You can have multiple files open at the same time: If you use see(file1) and then see(file2), you will have 2 open files at the same time, and if you now call get_char, it will read from file2. If you call see(file1) again, file1 will be active again. You can close these files in any order you want.

The read predicate is for parsing prolog terms. (It can handle DCG grammars as well.) You might need it for self-modifying programs, or for programs which transforms prolog code.

Reading a file the ISO style[edit]

When following the ISO standard, data from files and from network are handled the same way: as data streams.


	Open a file for reading: open(Filename, read, Streamhandler)

	Open a file for writing: open(Filename, write, Streamhandler)








process(File) :-
        open(File, read, In),
        get_char(In, Char1),
        process_stream(Char1, In),
        close(In).

process_stream(end_of_file, _) :- !.
process_stream(Char, In) :-
        print(Char),
        get_char(In, Char2),
        process_stream(Char2, In).

Example binary search tree with integer keys only





Prolog's built-in lists are handy, but sometimes a simple linear list is not enough. When you want to maintain an association between one set of elements ('keys') and another ('values'), you need an associative map data structure. Here's how such a structure can be implemented in Prolog using binary search trees (BSTs).

First we need a representation of search trees in Prolog terms. Remember that a BST is a binary tree with key-value pairs stored in the nodes. We can thus represent nodes as functors t(Key,Value,LeftChild,RightChild). The empty tree will be represented by the atom nil.

But we don't want to program to this representation directly: by providing the appropriate predicates, we can specify an abstract data structure (ADT). The reason for doing so is that we can later change our implementation to use balanced binary trees or some other more sophisticated data structure. We shall call this ADT the ordered map, or ordmap. Here are the basic operations on an ordmap:


	empty_map(-Map): unifies Map with the empty map.

	lookup_ordmap(+Key, +Map, -Value): lookup the Value associated with Key in Map.

	insert_ordmap(+Key, +Value, +Map0, -Map): insert the pair Key, Value in the map Map0, yielding Map. Note that afterwards, both Map0 and Map are valid ordered maps, which differ by at most one element. This predicate fails if Key is already present in Map0.

	update_ordmap(+Key, +Value, +Map0, -Map): like insert_ordmap, but removes any previous association of Key with a different value (and always succeeds).

	remove_ordmap(+Key, +Map0, -Map, -Value): remove Key from Map0, yielding Map. The value associated with Key is returned in Value. This predicate fails if Key was not in Map0.

	member_ordmap(+Map, -Key, -Value): backtracks over all key/value pairs in Map by order of keys.

	rmember_ordmap(+Map, -Key, -Value): backtracks over all key/value pairs in Map by reverse order of keys.

	size_ordmap(+Map, -Size): determines the number of elements in Map.



For reasons that will become clear later on, keys in our ordered maps should always be ground terms.

The implementation of empty_map is trivial:



empty_map(nil).



Our next predicate, lookup_map, follows the usual recursion patterns for BST operations:



lookup_ordmap(K, t(X,Y,L,R), V) :-
    (K == X ->
        V = Y
    ; K @< X ->
        lookup_ordmap(K,L,V)
    ;
        lookup_ordmap(K,R,V)
    ).



Note the use of ==/2 instead of unification. The reason for doing so lies in the use @</2 which compares terms according to the standard order of terms. In this ordering, for any two distinct terms [image: T_1] and [image: T_2], either [image: T_1] == [image: T_2], [image: T_1] @< [image: T_2], or [image: T_2] @< [image: T_1]. For example, a @< b, X @< Y and X @< foo. In fact, a free variable is always @< a ground term. But when two variables, or a variable and a ground term are unified, the ordering changes: after X=Y, X==Y is also true. This is why keys should, in principle, always be ground terms: that way, the ordering is always preserved (but we leave the appropriate check up to the user of our ordered map data structure). Note that there is no such restriction on values, since they don't need to be ordered.

Also note that we have no case for nil, since looking up anything in an empty tree will always fail.

Deletion of an element from a binary search tree can be a bit tricky to implement; the following code replaces a node with two children by its in-order predecessor, which is the maximum element of the left subtree. It uses the rm_max helper predicate to remove the maximum element from a subtree.



remove_ordmap(K, t(X,Y,L0,R), t(X,Y,L,R), V) :-
    K @< X,
    remove_ordmap(K,L0,L,V).
remove_ordmap(K, t(X,Y,L,R0), t(X,Y,L,R), V) :-
    K @> X,
    remove_ordmap(K,R0,R,V).
remove_ordmap(K, t(X,V,L,R), T, V) :-
    K == X,
    (L == nil ->
        T = R
    ; R == nil ->
        T = L
    ;
        rm_max(L,L1,K1,V1),
        T = t(K1,V1,L1,R)
    ).
rm_max(t(K,V,L,nil), L, K, V) :- !.
rm_max(t(X,Y,L,R0), t(X,Y,L,R), K, V) :-
    rm_max(R0,R,K,V).



The rest of the predicates are now easy to write:



insert_ordmap(K, V, nil, t(K,V,nil,nil)).
insert_ordmap(K, V, t(X,Y,L0,R), t(X,Y,L,R)) :-
    K @< X,
    insert_ordmap(K,V,L0,L).
insert_ordmap(K, V, t(X,Y,L,R0), t(X,Y,L,R)) :-
    K @> X,
    insert_ordmap(K,V,R0,R).

member_ordmap(t(X,Y,L,R), K, V) :-
    member_ordmap(L,K,V) ;
    (X=K, Y=V) ;
    member_ordmap(R,K,V).

size_ordmap(nil, 0).
size_ordmap(t(_,_,L,R), N) :-
    size_ordmap(L,NL),
    size_ordmap(R,NR),
    N is NL+NR+1.



Exercise: implement rmember_ordmap and update_ordmap.

Libraries[edit]

Associative map data structures are built into the libraries of various Prolog implementations (though often not in compatible ways):


	SICStus provides library(avl)

	SWI-Prolog provides library(assoc)



Both libraries are based on AVL trees; note that SICStus also provides a library(assoc), but that is based on simple lists.

Prolog's default search algorithm is depth-first search (DFS), which is sometimes not convenient: DFS doesn't always find the solution for all problems, and when it does, it might not find the optimal solution.

Let's say we want to find the shortest path through a directed graph with nodes represented by symbols and arc by the predicate arc/2. Then we can easily implement iterative deepening search to find the shortest path:



path(X, Z, Path) :-
    length(Path, _),
    path_r(X, Z, Path).

path_r(Z, Z, []).
path_r(X, Z, [X|Path]) :-
    arc(X, Y),
    path(Y, Z, Path).



path/3 backtracks over paths, ordered by length, if at least one path exists. Let's try it out:



?- path(a, g, P).
P = [a] ;
P = [a, b, f] ;
P = [a, c, d, f] ;
P = [a, b, c, d, f]



How this works: when length/2 is called with a variable first argument, it generates a list of the desired length containing fresh variables. When, in addition, the second argument is a variable, it backtracks over all possible lengths starting at zero and stepping by one. The helper predicate path_r is called each time with a fixed path length.

Sorting is a fundamental task in many programs. We explain how to write an implementation of the fast and general merge sort algorithm in Prolog.

The basic algorithm[edit]

The merge sort algorithm looks as follows:


function merge_sort(m)
    if length(m) ≤ 1
        return m
    else
        left, right = split_in_half(m)
        left = merge_sort(left)
        right = merge_sort(right)
        result = merge(left, right)
        return result


The first thing we have to do to implement any algorithm presented in this style in Prolog is to translate it from the language of functions to the language of predicates. Recall that Prolog has no return construct, but we can use an unbound variable as an output mechanism. So, we will write a predicate mergesort(Xs, S) that, given a list Xs, "returns" a sorted variant by binding S to it.

How about the if-then-else construct in the algorithm? We can translate that directly, but we also have the option of using Prolog's pattern matching. Because the conditional checks for ≤1, we have three structural cases: the empty list, a list of one element and the catch-all case. So, let's write our first merge sort.


% the empty list must be "returned" as-is
mergesort([], []).
% same for single element lists
mergesort([X], [X]).
% and now for the catch-all:
mergesort([X|Xs], S) :-
    length(Xs, Len),
    0 < Len,
    split_in_half([X|Xs], Ys, Zs),
    mergesort(Ys, SY),
    mergesort(Zs, SZ),
    merge(SY, SZ, S).


There's the skeleton of the algorithm. We have left out the definition of the helper predicates split_in_half and merge so far, but for a functioning sorting predicate, we must of course define those as well. Let's start with merge, which takes two sorted lists and produces a new sorted list containing all the elements in them. We use Prolog's recursion and pattern matching to implement merge.


% First two cases: merging any list Xs with an empty list yields Xs
merge([], Xs, Xs).
merge(Xs, [], Xs).
% Other cases: the @=< predicate compares terms by the "standard order"
merge([X|Xs], [Y|Ys], [X|S]) :-
    X @=< Y,
    merge(Xs, [Y|Ys], S).
merge([X|Xs], [Y|Ys], [Y|S]) :-
    Y @=< X,
    merge([X|Xs], Ys, S).


This predicate works, but it's repetitive and not very efficient. We'll revisit it in a minute, but first we must define split_in_half, which splits a list into two lists of roughly equal size (roughly because it may have an odd number of elements). To do so, we need the length of a list, which we unfortunately do not get as input (see merge_sort), so we need to compute that. We use a helper predicate split_at to actually split the list after computing the length.


split_in_half(Xs, Ys, Zs) :-
    length(Xs, Len),
    Half is Len // 2,    % // denotes integer division, rounding down
    split_at(Xs, Half, Ys, Zs).


Exercise. Before continuing to the definition of split_at, try to implement it yourself. Use recursion and case matching on both the input list and the second, length, argument.

Okay, now for split_at:


% split_at(Xs, N, Ys, Zs) divides Xs into a list Ys of length N
% and a list Zs containing the part after the first N.
split_at(Xs, N, Ys, Zs) :-
    length(Ys, N),
    append(Ys, Zs, Xs).


If the definition of split_at seems magical, then try to read it declaratively: "a list Xs, split after its first N elements into Ys and Zs, means that Ys has length N and Ys and Zs can be appended to retrieve Xs." The fact that this works is an example of Prolog's "bidirectional" power: many predicates can be run in "reverse order" to get the inverse of a computation.

Cleaning up[edit]

We now have a merge sort program that will sort any list, but some parts of it aren't particularly elegant. Let's first revisit merge_sort. We translated an if-then-else construct to multiple clauses since that's so common in Prolog, but we really had no reason to do so: the algorithm is deterministic, so there should no backtracking going on. Let's try a more direct rewrite of the pseudocode using Prolog's own if-then-else.


mergesort(Xs, S) :-
    length(Xs, Len),
    (Len =< 2 ->
        S = Xs
    ;
        split_in_half(Xs, Ys, Zs),
        mergesort(Ys, SY),
        mergesort(Zs, SZ),
        merge(SY, SZ, S)
    ).


Now, let's tackle merge's efficiency and readability, as promised. The last two cases of this predicate look like a copy-paste, which is always a code smell. Also, there may be unnecessary backtracking going on. Let's rewrite merge using an if-then-else as well:


% First two cases: merging any list Xs with an empty list yields Xs
merge([], Xs, Xs).
merge(Xs, [], Xs).
% Other cases: the @=< predicate compares terms by the "standard order"
merge(Xs, Ys, S) :-
    Xs = [X|Xs0],
    Ys = [Y|Ys0],
    (X @=< Y ->
        S = [X|S0],
        merge(Xs0, Ys, S0)
    ;
        S = [Y|S0],
        merge(Xs, Ys0, S0)
    ).

XPCE is the so-called Prolog graphic library , although it isn't Prolog at all. Like they mention in it's manual, XPCE is not Prolog; XPCE is an object-oriented language that has absolutely nothing to do with Prolog.

The first thing that we need to do to create applications in XPCE is to install Swi-Prolog. You can find it at it's distributor page.

The page about XPCE is http://www.swi-prolog.org/packages/xpce/.

to be continued...

Introduction[edit]

Higher-order programming means writing programs that take other programs as input and/or return still other programs as output. Higher-order programming is common in functional programming, but can be in done in Prolog as well.

As an example, suppose you are writing a speech synthesis application that reads out phone numbers aloud. We might have a predicate digit_english/2 that relates Prolog integers 0 through 9 to English words (as symbols):



digit_english(0,zero).
digit_english(1,one).
digit_english(2,two).
...



Now, to translate a phone number, define:



phone_english([],[]).
phone_english([Digit|Ds],[Word|Ws]) :-
    digit_english(Digit,Word),
    phone_english(Ds,Ws).



Now, suppose we want to add support for another language, say, German. You define digit_german:



digit_german(0,null).
digit_german(1,eins).
digit_german(2,zwei).
...



To translate a phone number, use phone_german:



phone_german([],[]).
phone_german([Digit|Ds],[Word|Ws]) :-
    digit_german(Digit,Word),
    phone_german(Ds,Ws).



Note the common recursion pattern in phone_english/2 and phone_german. You've probably seen that many times before, even written it several times. It can be summed up as:


	Base case: relate the empty list to itself.

	Recursive case: relate the first element of the first list to the first element of the second via some predicate, then recur on the tails of both lists.



This pattern can be captured by a higher-order predicate generally known as map/3 (but not defined by standard Prolog):



map([],P,[]).
map([X|Xs],P,[Y|Ys]) :-
    Goal =.. [P,X,Y],
    call(Goal),
    map(Xs,P,Ys).



In Prolog, we cannot write P(X,Y) since the head of a functor must be a symbol. Therefore, we use the infix operator =.. that unifies its first argument with a term built from its right argument, which must be a list. The first element, which must be a symbol, becomes the functor, the rest its arguments. We then apply call/1 to the term we constructed to invoke it as a Prolog goal.

We can now redefine phone_english/2 and phone_german/2 as:



phone_english(P,E) :- map(P,digit_english,E).
phone_german(P,G) :- map(P,digit_german,G).



Both predicates still work in both directions and non-determinism is preserved.

The befenit is obvious: by using map/3, we avoid repeating code fragments so our programs become shorter and easier to read and understand.

See also[edit]

The following research paper is about higher order programming in prolog and contains numerous examples: [1]

We've seen that in Prolog, a variable can be either bound (have a value, possibly another variable) or free (have no value). Constraint logic programming (CLP) extends the notion of a logical variable by allowing variables to have a domain rather than a specific value. Variables can also be constrained, which means that their value must abide by certain rules specified by the programmer. Constraint logic programming makes it possible to solve complex combinatorial problems with a minimum amount of code.

Introduction: the dif/2 constraint[edit]

Often, Prolog programming revolves around constraints on the values of variables, embodied in the notion of unification. For example, the goal X = f(1) can be said to constrain the variable X to take on the value f(1). Unification is of course more general than this, since X = Y constrains two variables to take on the same value, while X = f(_) constrains X in yet a third way: it may take on any of an infinite set of values that includes f([]) and f(hamburger), but the value it takes on must match the "template" f(_).

Standard unification is limited, though, in that it can only put a "positive" constraint on a variable. If we want to state that X may be bound to any value other than those matching f(_), then we can use unification together with negation, but we have to be very careful. As we've seen in an earlier chapter, we can sometimes use negation as failure to express inequality, but this is not always reliable:



?- X = 1, \+ X == 1.
false.

?- \+ X == 1, X = 1.
true



This non-monotonic property of \+ forces us to reorder goals to get variables bound at just the right time, making code harder to read, write and understand.


	Exercise. Write a predicate that succeeds when its input is a list containing only unique values. E.g. unique([1,2]) should succeed, but unique([foo,foo,1]) should fail because foo appears twice. How does your predicate deal with unique([X,X])? How about unique([A,B,C,D])?



Enter the dif predicate. This predicate, available in SWI and SICStus (but not strictly standard Prolog) allows us to constrain two variables to be different. The following SWI-Prolog session shows its operation:



?- dif(X,Y).
dif(X,Y).

?- dif(X, Y), member(X, [foo, bar]), member(Y, [foo, bar]).
X = foo,
Y = bar ;
X = bar,
Y = foo ;
false.



Contrast this with the result of the other inequality predicates \= and \==.


	Exercise. Implement the unique/1 predicate again, but this time use dif/2. What is Prolog's response to the query unique([A,B,C,D,E])? How many constraints does it report?




	Exercise. Write a predicate not_f1/1 that constrains its argument to not be of the form f(_), i.e. not a one-argument term with functor f Hint: use the =.. ("univ") operator.



Finite-domain constraints[edit]

We now turn to a much more powerful constraint handling system: constraint logic programming on finite domains, also known as CLP(fd). To use this, we must load a library. In SWI-Prolog, that's:



?- use_module(library(clpfd)).



Do you remember how arithmetic in Prolog fails when used with insufficiently instantiated variables? Try this for a starter:



?- X > Y, member(X,[1,2,3]), Y=2.



Even though there is logically one possible answer to this query (X=3), Prolog cannot infer it. We can get the right answer by rearranging the conjuncts of the query:



?- member(X,[1,2,3]), Y=2, X > Y.



CLP(fd) is a lot smarter than Prolog when it comes to arithmetic. Try this (note: ECLiPSe users should replace in by ::):



?- X #> Y, X in 1..3, Y=2.



This succeeds with the only correct answer, X=3. The predicate #>/2, when given two variables, posts the constraint that it left argument should be greater than its right argument. CLP(fd) holds this constraint in its constraint store until it knows enough about the variables' possible values to infer further information from it. The moment that Y gets the value 2, CLP(fd) infers that X can only have the value 3.

What happens when we bind Y to 1 instead?



?- X #> Y, X in 1..3, Y=1.
Y = 1,
X in 2..3.



Rather than start backtracking over the possible values of X, CLP(fd) constrains X to a smaller domain and stops. To have CLP(fd) search for possible assignments to variables, we must tell it do so explicitly. The simplest way to do that is with the predicate label/1:



?- X #> Y, X in 1..3, Y=1, label([X]).
X = 2,
Y = 1 ;
X = 3,
Y = 1.



What we've seen so far isn't very interesting, because there was only one constraint variable. Let's have a look at a problem with eight variables.

Send more money[edit]

The send more money puzzle is the quintessential example of a constraint problem. It amount to assigning different digits 0 through 9 to the variables [S,E,N,D,M,O,R,Y] such that the sum


   SEND
+  MORE
= MONEY


is solved; S and M should both be greater than zero. Instead of typing to the Prolog prompt, let's make a proper Prolog module.



:- use_module(library(clpfd)).

sendmoremoney(Vars) :-
    Vars = [S,E,N,D,M,O,R,Y],
    Vars ins 0..9,
    S #\= 0,
    M #\= 0,
    all_different(Vars),
                 1000*S + 100*E + 10*N + D
    +            1000*M + 100*O + 10*R + E
    #= 10000*M + 1000*O + 100*N + 10*E + Y.



ins/2 is the same as in/2, except that it sets the domains of several variables at the same time. (ECLiPSe: use ::, SICStus: use domain(Vars,0,9)). all_different (which may be called all_distinct in some implementations) is logically equivalent to posting disequality constraints (#\=) on all pairs of variables in Vars, but is shorter and possibly more efficient. Note that we can express the sum as one constraint over all eight variables at the same time. Let's try this out:



?- sendmoremoney([S,E,N,D,M,O,R,Y]).
S = 9,
M = 1,
O = 0,
E in 4..7,
all_different([E, N, D, R, Y, 0, 1, 9]),
1000*9+91*E+ -90*N+D+ -9000*1+ -900*0+10*R+ -1*Y#=0,
N in 5..8,
D in 2..8,
R in 2..8,
Y in 2..8.



Even without labeling variables, CLP(fd) has inferred the values of three variables, has simplified the sum and put tighter bounds on the remaining five variables. Ofcourse, we want values for all our variables:



?- Vars=[S,E,N,D,M,O,R,Y], sendmoremoney(Vars), label(Vars).
Vars = [9, 5, 6, 7, 1, 0, 8, 2],
S = 9,
E = 5,
N = 6,
D = 7,
M = 1,
O = 0,
R = 8,
Y = 2



This should run instantly. Exercise: implement the puzzle in Prolog without CLP(fd), and notice how long it takes to run.

Prolog has a mechanism for defining certain grammars called Definite Clause Grammar notation. This makes it easy to write parsers. Note that while DCG syntax is part of the ISO standard, DCG semantics are not.[citation needed]

Grammar rules can be written in the form:


head --> body


For example:


sentence --> noun_phrase, verb_phrase.


Which will be translated to:


sentence(L0,LREMAINDER):-
   noun_phrase(L0,L1),verb_phrase(L1,LREMAINDER).


It means, that the sentence clause will receive L0 as input, and after parsing from a sentence from L0, it will give back LREMAINDER. Let's assume that your start symbol is sentence. Then LREMAINDER is expected to be [ ] after a succesful parsing. The interpretation of the body of this clause is: if we parse a noun_phrase and a verb_phrase from our sentence, we will get back an empty list.

You can also call prolog predicates using braces.

Example[edit]

An example DCG program, which can parse numbers:


number --> digit, number_remaining.
number_remaining --> dot,number_remaining.
number_remaining --> digit,number_remaining.
number_remaining([],[]).
dot -->[0'.].
digit --> [J], {digit_code(J)}.
digit_code(J):- J >= 0'0, J =< 0'9.


Note: 0'9 means the character 9 (or more precisely, the character code of 9, because there's no distinct character datatype in SWI).

If you try to consult this program, you might get a warning, because we redefined the built-in number/2 predicate.

Running the program:


?- number("120",L).
L = [] ;
fail.


We get back one "result": it means that the parsing wasn't ambiguous. (There's only one possible parsing tree of the input.)

One of the earlier examples, revappend, can be written using DCG:


revappend([]) --> [].
revappend([X|Xs]) --> revappend(Xs), [X].


External links[edit]

Chomsky hierarchy of grammars on Wikipedia

Prolog programs are rarely used standalone, they are usually embedded in imperative programming languages like in Java, C or C++. It's usually easier to embed it in languages, which support OOP than in C, because the Prolog types are very different from the ones used in imperative languages, and objects can encapsulate their specific behaviour quite well. (e.g. the "not instantiated" state of a variable)

SWI-Prolog[edit]

You can find the documentation about Java-Prolog interface here: [1]

Interfacing with Perl: [2]

Other Java-Prolog links[edit]

[3]

